' Demonstration program illustrating the use of the Hitachi

' H48C 3-Axis accelerometer module from Parallax.  This code

' is based on the example code that is provided with the module.

'

Dio      PIN 15      ' Pin 2 on the H48C Module

Clk      PIN 14      ' Pin 1 on the H48C Module

CS       PIN 13      ' Pin 5 on the H48C Module

Cnt2Mv   CON 52812   ' Conversion factor: 12bit count to millivolts

GfCnv    CON 14418   ' Conversion factor: 12bit count to mG (accel.)

Vref     CON 3

axis     VAR Nib     ' axis counter

XAxis    VAR Word    ' X-Axis acceleration, in milli-g's

YAxis    VAR Word    ' Y-Axis acceleration, in milli-g's

ZAxis    VAR Word    ' Z-Axis acceleration, in milli-g's

rvCount  VAR Word    ' Reference voltage counter - 12bit ADC result

axCount  VAR Word    ' axis voltage counter - 12bit ADC result

mVolts   VAR Word    ' temp variable for milli-volts

gForce   VAR Word    ' temp variable for acceleration

Reset:      ' Clear debug Screen

  HIGH CS

  DEBUG CLS,

        "      Count  Volts   mG     G  ", CR,

        "      -----  -----  ----  -----", CR,

        "   X                           ", CR,

        "   Y                           ", CR,

        "   Z                           ", CR,

        "      Vref                     "

Main:

  axis = 0            ' Read X-Axis data and display result

  GOSUB Read_H48C

  XAxis = gForce

  GOSUB display

  axis = 1            ' Read Y-Axis data and display result

  GOSUB Read_H48C

  YAxis = gForce

  GOSUB display

  axis = 2            ' Read Z-Axis data and display result

  GOSUB Read_H48C

  ZAxis = gForce

  GOSUB Display

  GOSUB Orientation   ' Determing Robot orientation

  PAUSE 20

GOTO main

Read_H48C:            ' Read the H48C module

  LOW CS              ' 1st read in a new reference voltage

  SHIFTOUT Dio, Clk, MSBFIRST, [%11\2, VRef\3]

  SHIFTIN  Dio, Clk, MSBPOST, [rvCount\13]

  HIGH CS

  PAUSE 1

  LOW CS              ' 2nd read in voltage for axis of interest

  SHIFTOUT Dio, Clk, MSBFIRST, [%11\2, axis\3]

  SHIFTIN  Dio, Clk, MSBPOST, [axCount\13]

  HIGH CS             ' Convert result into milli-g's

  IF (axCount >= rvCount) THEN

      gForce = (axCount - rvCount) ** GfCnv

    ELSE

      gForce = -((rvCount - axCount) ** GfCnv)

  ENDIF

RETURN

Display:              ' Display measurements for analysis

    DEBUG CRSRXY, 6, (2+axis), DEC axCount

    mVolts = axCount ** Cnt2Mv

    DEBUG CRSRXY, 13, (2+axis),

          DEC (mVolts/1000), ".",

          DEC3 mVolts//1000

    DEBUG CRSRXY, 20, (2+axis),

          " " + (gForce.BIT15 * 13),

          DEC3 ABS(gForce)

    DEBUG CRSRXY, 26, (2+axis),

          " " + (gForce.BIT15 * 13),

          DEC1 (ABS(gForce)/100), ".",

          DEC2 ABS(gForce)//100

    mVolts = rvCount ** Cnt2Mv

    DEBUG CRSRXY, 13, 5,

          DEC(mVolts/1000), ".",

          DEC3 mVolts//1000

  RETURN

Orientation:          ' Display basic robot orientation

  IF ZAxis.BIT15 = 0 THEN

    DEBUG CRSRXY, 2, 7, "Robot is vertical "

  ELSE

    DEBUG CRSRXY, 2, 7, "Robot is inverted "

  ENDIF

  IF YAxis.BIT15 = 1 THEN

    DEBUG CRSRXY, 2, 8, "Robot is leaning forward   "

  ELSE

    DEBUG CRSRXY, 2, 8, "Robot is leaning backwards "

  ENDIF

  IF Xaxis.BIT15 = 0 THEN

    DEBUG CRSRXY, 2, 9, "Robot is leaning towards the port side     "

  ELSE

    DEBUG CRSRXY, 2, 9, "Robot is leaning towards the starbored side"

  ENDIF

RETURN
